
極形式グラフィックスによるいろいろな渦巻き曲線(Spiral)

J-plotによるグラフィック・プログラム

西川　利男

０．はじめに－極形式は特殊ではない

　グラフィックスで図形を表示するには、線を点の集合とみて、その位置をタテヨ

コXY座標を用いた直交形式で行うことが多い。

　しかしながら、自然界では、原子分子の内部から、宇宙空間に至るまで、極座標の

方がかえって自然であり、極形式による表示は直交形式に比べて、決して特殊もので

はない。

　極座標表示とは、ある点から見上げる角度θとその方向の動径の長さrとの関係式

として示される。しかし、コンピュータグラフィックスではふつう直交座標(x, y)で

行われるので、極座標(r、θ)からはつぎのような変換が必要である。
　　  sin,cos ryrx 

　この後、扱おうとする渦巻き曲線、らせん(Spiral)は極形式グラフィックスにより、

極めて自然に効率的に行われる。

１．いろいろならせん＝渦巻き(Spiral)曲線[1,2]とそのグラフィック・プログラム

[1] 岩波数学公式Ｉp.285　

[2] Wikipedia,  渦巻、代数螺旋、など

１．１アルキメデスのらせん

　もっとも基本的で単純であるが、

いろいろなところに現れる。　　

　　  ar  … *)　aは定数

　　　　

  Jのプログラムは次のようになる。

load 'trig'

load 'plot' 

NB. Archimedes' Spiral

alch =: 3 : 0

n =. y.

if. 0 = #n do. n =. 48 end.

th=. (10 * i. >: n) * 1r120p1

a =. 1

r =. a * th

plot (r * cos th); (r * sin th)

)

*)  ふつう、このような式で表されることが多いが、極座標とはrとθとの２つの値

で表すことなので、個人的には(r、θ)＝( a 、θ)  とするのが良いと思う。

１．２　代数らせん、

　一般に動径の長さaが角度θの代数関数で表されるものが代数らせんと呼ばれる。

　
kar 

　

１．２．１　放物らせん

特に、k = 1/2のもの

　  ar

は、放物らせんであり、Jのプログラムは次のようになる。

NB. Parabolic Spiral

parab =: 3 : 0

n =. y.

if. 0 = #n do. n =. 198 end.

th=. (10 * i.>: n) * 1r120p1

a =. 1

r =. a * %: th

plot (r * cos th); (r * sin th)

) 

  放物らせんは巻き数を増やすとだんだん詰まってくる。それに対してアルキメデス

のらせんは巻き数を増やしても一定である。

１．２．２　双曲らせん

　双曲らせん、k=-1 は次の式で表される。

　 
1
 ar

１．２．３　ラッパ線（リチウス）

　k = - ½　はリチウスと呼ばれて、特異な形を示す。渦を巻いて原点(r = 0)に近づ

いていく。

　 
1

 ar

１．３　対数らせん

　
ar 

Jのプログラムは次のようになる。

NB. Logarithmic Spiral

logspiral =: 3 : 0

1.5 logspiral ''

:

th=. (10 * i.49) * 1r120p1

a =. x.

NB. a =. 1.5

r =. a ^ th

plot (r * cos th); (r * sin th)

)

  パラメータを変えると、いろいろならせんが現れる。

　1.2 logspiral 180

          1.5 logspiral 180

プログラム・リスト

NB. Spiral Graphics in Polar Expression

NB.   by T. Nishikawa

NB. 岩波数学公式Ｉp.285　

load 'trig'

load 'plot' 

NB. Archimedes' Spiral

alch =: 3 : 0

th=. (10 * i.49) * 1r120p1

a =. 1

r =. a * th

plot (r * cos th); (r * sin th)

)

NB. Parabolic Spiral

parab =: 3 : 0

n =. y.

if. 0 = #n do. n =. 199 end.

th=. (10 * i.n) * 1r120p1

a =. 1

r =. a * %: th

plot (r * cos th); (r * sin th)

) 

NB. Hyperbolic Spiral

hyperb =: 3 : 0

n =. y.

if. 0 = #n do. n =. 99 end.

th=. (10 * >: i.n) * 1r120p1

a =. 0.1

r =. a * % th

plot (r * cos th); (r * sin th)

) 

NB. Lithius Spiral

lith =: 3 : 0

n =. y.

if. 0 = #n do. n =. 199 end.

th=. (10 * >: i.n) * 1r120p1

a =. 0.1

r =. a * % %: th

plot (r * cos th); (r * sin th)

)

NB. Logarithmic Spiral

logspiral =: 3 : 0

1.5 logspiral ''

:

n =. y.

if. 0 = #n do. n =. 198 end.

th=. (10 * i. >: n) * 1r120p1

a =. x.

NB. a =. 1.5

r =. a ^ th

plot (r * cos th); (r * sin th)

)

